

Tema 13

TEME:

- Cvetići od listića u boji napravljeni kroz procedure

PRIMER 1: Crtanje listića uz pomoć parametara (od listica cemo formirati cvetice)

Korak 1: Slika je simetrična, nacrtacemo polovinu, pa onda ponoviti to 2 puta. To “nesto” je jedan luk i okretaj kornjace



Korak 2: Crtanje jednog luka (npr trecina kruga, $360/3 = 120$)

```
repeat 120 [fd 2 rt 1]
```



Korak 3: okretaj kornjače kod šiljka je za 60 ($360^\circ - 120^\circ - 120^\circ = 120^\circ$ i $120^\circ/2 = 60^\circ$)

```
repeat 120 [fd 2 rt 1] rt 60
```

Korak 4: ponavljanje 2 puta radi crtanja celog lista

```
repeat 2 [repeat 120 [fd 2 rt 1] rt 60]
```

Korak 5: Procedura sa parametrom za podesavanje sirine listica (gde su delovi A i B kraći) ili šire lišće (gde su delovi A i B duži).



Korisićemo proceduuru sa parametrom :UGAO. Taj ugao za okretanje kornjače na šiljku lista se može računati sa $180^\circ - :UGAO$.


```
to LISTIC :UGAO
```

```
repeat 2 [repeat :UGAO [fd 2 rt 1] rt 180-:UGAO]
```

```
end
```

Korak 6: Testirati

```
listic 20  
listic 40  
listic 60  
listic 80  
listic 100
```



PRIMER 2: Crtanje cveta uz pomoć listica sa parametrom

Korak 1: koristimo proceduru za listic iz prethodnog primera

```
to LISTIC :UGAO
repeat 2 [repeat :UGAO [fd 2 rt 1] rt 180-:UGAO]
end
```

Korak 2: Formiranje cveta sa proizvoljnim uglom. Imati na umu da proizvod od ostala dva broja - broja ponavljanja i rotacije izmedju dva listica treba da bude 360.

```
repeat 18[listic 100 rt 20]
```

Korak 3: Napraviti jos nekoliko cvetova razlicitog broja latica.

npr:


```
repeat 36[listic 60 rt 10]
```

OBJASNJENJE:

- Napraviti sve faktore broja 360 i onda povezati koji parovi mogu dati kakve krugove.

PRIMER 3: Dodavanje boje na prethodni cvet.

- Tako da svaka latica bude razlicite boje
- Tako da svaka latica bude iste boje



Korak 1: a) Dodavanje boje unutar procedure

```
to LISTIC :UGAO
  setcolor random 15
  repeat 2 [repeat :UGAO [fd 2 rt 1] rt 180-:UGAO]
end
```

```
repeat 36[listic 60 rt 10]
```

Korak 2: b) Dodavanje boje pre poziva procedure


```
to LISTIC :UGAO
  repeat 2 [repeat :UGAO [fd 2 rt 1] rt 180-:UGAO]
end
setcolor random 15
repeat 36[listic 60 rt 10]
```

PRIMER 4: Napiši proceduru LATICA :UGAO :VELICINA sa dva parametra :UGAO i :VELICINA, tako da možemo podesiti dužinu luka i veličinu lišća.

Napraviti crtanje takva 3 cveta, razlicite velicine i boje.

Korak 1: Dodajemo parametar za velicinu

```
to LATICA :ugao :velicina
  repeat 2 [repeat :ugao [fd :velicina rt 1] rt
180-:ugao]
end
```



Korak 2: Poziv procedure radi testiranja.

```
LATICA 60 3
```

Korak 3: Dodavanje crtanja cveta

```
to LATICA :ugao :velicina
repeat 2 [repeat :ugao [fd :velicina rt 1] rt 180-:ugao]
end
setcolor random 15
repeat 18[LATICA 60 3 rt 20]
```

Korak 4: Dodavanje toga u repeat 3 puta

```
to LATICA :ugao :velicina
repeat 2 [repeat :ugao [fd :velicina rt 1] rt 180-:ugao]
end
```

```
repeat 3 [
  setcolor random 15
  repeat 18[LATICA 60 3 rt 20]
]
```

Korak 5: pomeranje da se crtaju sva 3 na razlicitim pozicijama (i vel cveta je smanjena)

```
to LATICA :ugao :velicina
repeat 2 [repeat :ugao [fd :velicina rt 1] rt 180-:ugao]
end
repeat 3 [
  setcolor random 15
  pu
  setx random 400
  sety random 400
  pd
  repeat 18[LATICA 60 1 rt 20]
]
```

Korak 6: randomizacija velicine

```
to LATICA :ugao :velicina
repeat 2 [repeat :ugao [fd :velicina rt 1] rt 180-:ugao]
end
repeat 3 [
  setcolor random 15
  pu
  setx random 400
  sety random 400
  make "vel random 5
  pd
  repeat 18[LATICA 60 :vel rt 20]
]
```